```
100 '************************
110 '**                    **
120 '** = SNIDER'S MAZE =  **
130 '**                    **
140 '**  (c)  Juli  1985   **
150 '**  Dietmar Schulze   **
160 '**  Katharinenhof 5   **
170 '**  5000  KOELN   1   **
180 '**                    **
190 '**  Tel.0221/326121   **
200 '**                    **
210 '**        fuer        **
220 '**                    **
230 '** SCHNEIDER CPC 464  **
240 '**                    **
250 '************************
260 '
270 '***  S N I D E R ' S   M A Z E  ***
280 '
290 GOSUB 5020'===> zur Initialisierung
300 '
500 '*** Eingabe Art und Groesse ***
510 '
```

```
520 wd=5:h=1:flag=2:RESTORE 7300:GOSUB 7090
530 PEN 1:efl=0:gl=0:CLS:CLS#2:PRINT:PRINT"  Soll Labyrinth erkennbar sein (J/N)
 ?";
540 a$="":WHILE a$<>"j"AND a$<>"n":a$=LOWER$(INKEY$):WEND:IF a$="n"THEN gl=-1
550 CLS:CLS#2:PEN 1:PRINT:PRINT"  Waehle max.= 38*19, min.= 5*5 Felder"
560 LOCATE 3,4:PRINT"Eingabe (z.B. 6,10) : "CHR$(18);:LINE INPUT xy$
570 a=INSTR(xy$,",")
580 ux=VAL(LEFT$(xy$,a-1)):uy=VAL(RIGHT$(xy$,LEN(xy$)-a)):IF ux<5 OR uy<5 OR ux>
38 OR uy>19 THEN PRINT#2," Noch einmal, bitte !!!";:FOR a=1 TO 3000:NEXT:CLS#2:G
OTO 560
590 '
600 '=== Neue Groesse dimensionieren ===
610 '
620 ERASE a,lzeil$
630 DIM a(ux+1,uy+1),lzeil$(uy+1)
640 '
1000 '*** Aufbau der vollen Zellen ***
1010 '
1020 CLS:RANDOMIZE TIME*ux*uy
1030 FOR y=0 TO uy+1:FOR x=0 TO ux+1:a(x,y)=16-SGN((x>0 AND x<=ux)*(y>0 AND y<=u
y)):LOCATE x+1,y+1:PRINT CHR$(a(x,y)+128);
1040 NEXT x,y:lzeil$(0)=" "+STRING$(ux-1,130)+CHR$(245)+" ":lzeil$(uy+1)=" "+CHR
$(245)+STRING$(ux-1,136)+" "
1050 FOR a=0 TO 1:LOCATE 1,1+a*(uy+1):PRINT lzeil$(a*(uy+1));:FOR b=2 TO uy+1:LO
CATE 1+a*(ux+1),b:PRINT CHR$(132-a*3);:NEXT b,a
1060 z=ux*uy:fz=0:PRINT#2,"  "DEC$((z,"###")" Felder gesamt,";:LOCATE#2,22,1:PRI
NT#2,DEC$((fz,"###")" sind fertig !";
1070 x=INT(RND*ux+1):y=INT(RND*uy)+1:flag=0
1080 '
1500 '*** Zellen aufbrechen mit RND ***
1510 '
1520 FOR a=0 TO 3:gef(a)=0:NEXT
1530 nri=INT(RND*4):GOSUB 4020:gef=gef(0)*gef(1)*gef(2)*gef(3):IF gef=0 THEN 153
0
1540 IF flag THEN 1820 ELSE IF fz>INT(z*0.9)THEN 1570
1550 x=INT(RND*ux+1):y=INT(RND*uy+1):IF a(x,y)>14 THEN 1550
1560 GOTO 1520
1570 IF x1=0 AND y1=0 THEN FOR x1=1 TO ux:FOR y1=1 TO uy
1580 x=x1:y=y1:IF gef>0 THEN 1600
1590 IF a(x,y)<15 THEN IF a(x+1,y)=15 OR a(x-1,y)=15 OR a(x,y+1)=15 OR a(x,y-1)=
15 THEN 1520
1600 gef=0:NEXT y1,x1:IF fz+1<z THEN x1=0:y1=0:GOTO 1570
1610 '
1800 '*** Labyrinth ist fertig ***
1810 '
1820 eg=ux:ag=1:a(ag,uy)=a(ag,uy)-2:a(eg,1)=a(eg,1)-8:PEN 3:LOCATE eg+1,1:PRINT
CHR$(245):LOCATE ag+1,uy+2:PRINT CHR$(245);
1830 FOR a=1 TO uy:lzeil$(a)=CHR$(132):FOR b=1 TO ux:lzeil$(a)=lzeil$(a)+CHR$(a(
b,a)+128):NEXT:lzeil$(a)=lzeil$(a)+CHR$(129):NEXT
1840 IF gl THEN FOR pa=1 TO 2500:NEXT:GOTO 2140
1850 PEN 1:LOCATE eg+1,2:PRINT CHR$(a(eg,1)+128);:LOCATE ag+1,uy+1:PRINT CHR$(a(
ag,uy)+128);
1860 FOR pa=1 TO 2500:NEXT:PRINT#2," Soll Mr.Snider Durchgang suchen (J/N)?";
1870 a$=LOWER$(INKEY$):IF NOT(a$="j"OR a$="n")THEN 1870 ELSE IF a$="n"THEN 2140
1880 '
2000 '*** Durchgangs-Suche Mr. Snider ***
2010 '
2020 x=eg:ax=x:y=1:ay=0:fz=1:ORIGIN 0,0
2030 CLS#2:PRINT#2,DEC$((fz-1,"#####")"  Schritte bis jetzt gebraucht !";:GOSUB
4520:PEN 3:LOCATE ux+1,1:PRINT CHR$(245);:PEN 1
2040 su=1
```

```
2050 xy=a(x,y):IF(xy AND 2^((su+1)MOD 4))=0 THEN su=(su+1)MOD 4:GOTO 2090
2060 IF(xy AND 2^su)=0 THEN 2090
2070 IF(xy AND 2^((su+3)MOD 4))=0 THEN su=(su+3)MOD 4:GOTO 2090
2080 su=(su+2)MOD 4
2090 fz=fz+1:x=x+sux(su):y=y+suy(su)
2100 GOSUB 4520'==> Zeichnen Mr.Snider
2110 IF y<uy+1 THEN 2050
2120 CLS#2:PRINT#2,DEC$((fz-1,"#####")"  Schritte brauchte Mr. Snider !";:wd=11:
RESTORE 7190:GOSUB 7090
2130 PEN 3:FOR pa=1 TO 15:FOR a=0 TO 2:LOCATE x+1,y+1:PRINT CHR$(144+a):FOR b=1
TO 200:NEXT b,a,pa
2140 PRINT#2," Willst DU den Durchgang suchen (J/N) ?";
2150 a$=LOWER$(INKEY$):IF a$="n"THEN 530 ELSE IF a$<>"j"THEN 2150
2160 CLS#2:PRINT#2,"     In diesem Labyrinth (J/N)  ?";
2170 a$=LOWER$(INKEY$):IF a$<>"j"AND a$<>"n"THEN 2170 ELSE IF a$="n"THEN 530
2180 '
3000 '*** Eigene Suche durch 3D-Maze ***
3010 '
3020 a(eg,1)=a(eg,1)+8:a(ag,uy+1)=7:ax=eg:ay=1:su=3:cf=1:CLS#2:ORIGIN 318,184:fz
=0:PRINT#2,"   0 Schritte";
3030 GOSUB 4220'==> 3D-Grafik-Aufbau
3040 fig=(fig+1)MOD 3:a$=INKEY$:IF a$=""THEN LOCATE 20,19:PRINT CHR$(144+fig);:F
OR pa=1 TO 150:NEXT:GOTO 3040
3050 a$="":IF INKEY(8)>-1 THEN suneu=(su+1)MOD 4:GOSUB 3110:GOTO 3040
3060 IF INKEY(1)>-1 THEN suneu=(su+3)MOD 4:GOSUB 3110:GOTO 3040
3070 IF INKEY(2)>-1 THEN suneu=(su+2)MOD 4:GOSUB 3110:GOTO 3040
3080 IF INKEY(0)>-1 THEN GOSUB 3130:ON efl+2 GOTO 3150,3040
3090 IF INKEY(9)>-1 THEN GOSUB 3200:GOTO 3040
3100 GOTO 3040
3110 cf=0:GOSUB 4230:IF flag THEN LOCATE 11,10:PRINT SPACE$(19):flag=0
3120 cf=1:su=suneu:GOSUB 4230:RETURN
3130 IF(a(ax,ay)AND 2^su)=0 THEN cf=0:GOSUB 4230:ax=ax+sux(su):ay=ay+suy(su):GOS
UB 3120:LOCATE#2,1,1:fz=fz+1:PRINT#2,DEC$((fz,"####");:IF ay=uy+1 THEN efl=-1:RE
TURN ELSE RETURN
3140 LOCATE 20,19:PRINT" ";:GOSUB 7040:FOR p1=1 TO 10:FOR fig=0 TO 2:LOCATE 20,1
5:PRINT CHR$(144+fig):FOR pa=1 TO 50:NEXT pa,fig,p1:LOCATE 20,15:PRINT" ":RETURN
3150 PRINT#2," Schritte in die Freiheit !!";
3160 wd=11:RESTORE 7190:GOSUB 7090
3170 FOR j=1 TO 3:FOR a=-15 TO 15:FOR b=0 TO 1:LOCATE 20,ABS(a)+4:PRINT CHR$(144
+b);:FOR pa=1 TO 20:NEXT:LOCATE 20,ABS(a)+4:PRINT" ";:NEXT b,a,j
3180 CLS#2:PRINT#2,"          Noch ein Versuch (J/N) ?"
3190 a$=LOWER$(INKEY$):IF a$="j"THEN 530 ELSE IF a$="n"THEN END ELSE 3190
3200 CLS:PEN 1:FOR a=0 TO uy+1:LOCATE 1,a+1:PRINT lzeil$(a);:NEXT
3210 PEN 3:FOR a=0 TO 1:LOCATE 2+(ux-1)*a,uy+2-(uy+1)*a:PRINT CHR$(245);:NEXT:LO
CATE ax+1,ay+1:PRINT CHR$(22)CHR$(1)CHR$(146)CHR$(22)CHR$(0);
3220 PAPER 2:PEN 1:LOCATE 9,21:PRINT" "CHR$(148+su)" ist Blickrichtung ! ";:PAPE
R 0
3230 LOCATE#2,1,1:PRINT#2,"          Zurueck mit einer Taste";:a$=INKEY$:WHILE INK
EY$="":WEND:CLS#2:PRINT#2,DEC$((fz,"####")" Schritte"CHR$(18);:GOSUB 4220:RETURN
3240 '
4000 '*** Subroutine Zellenausgabe ***
4010 '
4020 adx=0:ady=0:IF nri=0 THEN adx=-1 ELSE IF nri=2 THEN adx=1 ELSE IF nri=3 THE
N ady=-1 ELSE ady=1
4030 prf=a(x+adx,y+ady):IF prf<>15 THEN gef(nri)=1:RETURN
4040 a(x,y)=a(x,y)-2^nri
4050 GOSUB 4090:x=x+adx:y=y+ady:a(x,y)=a(x,y)-2^((nri+2)MOD 4)
4060 GOSUB 4090:fz=fz+1:LOCATE#2,22,1:PRINT#2,DEC$((fz+1,"###");
4070 IF fz+1=z THEN flag=1:FOR a=0 TO 3:gef(a)=1:NEXT
4080 FOR a=0 TO 3:gef(a)=0:NEXT:RETURN
4090 LOCATE 1+x,1+y:IF NOT gl THEN PRINT CHR$(a(x,y)+128)ELSE PRINT CHR$(146)
```

```
4100 RETURN
4110 '
4200 '*** 3D-Labyrinth-Darstellung ***
4210 '
4220 CLS:PEN 3:PRINT CHR$(22)CHR$(1);:FOR a=0 TO 1:LOCATE 4,2+a*18:PRINT STRING$
(34,216+a*2);:FOR b=2 TO 20:LOCATE 4+a*33,b:PRINT CHR$(219-a*2);:NEXT b,a:PRINT
CHR$(22)CHR$(0);
4230 zend=1:x=ax:y=ay
4240 IF(a(x,y)AND 2^su)=0 THEN zend=zend+1:x=x+sux(su):y=y+suy(su):GOTO 4240
4250 x=ax:y=ay:FOR b=1 TO zend:a=0:IF(a(x,y)AND 2^((su+3)MOD 4))=0 THEN gf=0 ELS
E gf=1
4260 GOSUB 4280:a=1:IF(a(x,y)AND 2^((su+1)MOD 4))=0 THEN gf=0 ELSE gf=1
4270 GOSUB 4280:x=x+sux(su):y=y+suy(su):NEXT b:RETURN
4280 PLOT gp(b,a,0),gp(b,a,1),cf:DRAW gp(b,a,2),gp(b,a,3)
4290 DRAW gp(b-1,a,2),gp(b-gf,a,3):PLOT gp(b,a,0),gp(b,a,1):DRAW gp(b-1,a,0),gp(
b-gf,a,1)
4300 IF zend=1 THEN IF a=1 THEN IF x=ux AND y=1 AND su=3 THEN LOCATE 11,10:PEN 2
:PRINT"Das war der Eingang":PEN 3:flag=-1
4310 IF y<>uy+1 AND b=zend AND a=1 THEN FOR c=0 TO 1:PLOT gp(b,c,0),gp(b,c,3):DR
AW gp(b,1-c,0),gp(b,c,3):NEXT
4320 RETURN
4330 '
4500 '=== Zeichnen Mr. Snider ===
4510 '
4520 LOCATE#2,1,1:PRINT#2,DEC$((fz,"#####");:PRINT CHR$(22)CHR$(1);:col=TEST(5+x
*16,399-53-y*16)MOD 3:GOSUB 4550
4530 PEN 0:FOR d=0 TO 1:LOCATE ax+1,ay+1:PRINT CHR$(233-d*86);:PEN 2-col:NEXT d
4540 ax=x:ay=y:GOSUB 4550:PEN 1:PRINT CHR$(22)CHR$(0);:RETURN
4550 FOR a=0 TO 1:PEN 0:LOCATE ax+1,ay+1:PRINT CHR$(233);:PEN 3:LOCATE ax+1,ay+1
:PRINT CHR$(144+a);:FOR pa=1 TO 50:NEXT pa,a:RETURN
4560 END
4570 '
5000 '*** Selbstdefinierte Zeichen ***
5010 '
5020 SYMBOL AFTER 128
5030 SYMBOL &80,&0,&0,&0,&0,&0,&0,&0,&0
5040 SYMBOL 129,128,128,128,128,128,128,128,128
5050 SYMBOL 130,0,0,0,0,0,0,0,255
5060 SYMBOL 131,128,128,128,128,128,128,128,255
5070 SYMBOL 132,1,1,1,1,1,1,1,1
5080 SYMBOL 133,129,129,129,129,129,129,129,129
5090 SYMBOL 134,1,1,1,1,1,1,1,255
5100 SYMBOL 135,129,129,129,129,129,129,129,255
5110 SYMBOL 136,255,0,0,0,0,0,0,0
5120 SYMBOL 137,255,128,128,128,128,128,128,128
5130 SYMBOL 138,255,0,0,0,0,0,0,255
5140 SYMBOL 139,255,128,128,128,128,128,128,255
5150 SYMBOL 140,255,1,1,1,1,1,1,1
5160 SYMBOL 141,255,129,129,129,129,129,129,129
5170 SYMBOL 142,255,1,1,1,1,1,1,255
5180 SYMBOL 143,255,129,129,129,129,129,129,255
5190 SYMBOL 144,0,56,40,124,60,104,12,0
5200 SYMBOL 145,&0,&38,&28,&7C,&78,&2C,&60,&0
5210 SYMBOL &92,&0,&38,&28,&7C,&7C,&28,&6C,&0
5220 SYMBOL 147,0,0,60,60,60,60,0,0
5230 SYMBOL 148,24,48,96,223,223,96,48,24
5240 SYMBOL 149,24,24,24,153,219,102,60,24
5250 SYMBOL 150,24,12,6,251,251,6,12,24
5260 SYMBOL 151,24,60,102,219,153,24,24,24
5270 '
6000 '*** Initialisierung ***
```

```
6010 '
6020 MODE 1:DEFINT a-z
6030 BORDER 13:INK 0,0:INK 1,23:INK 2,2:INK 3,15:PEN 1
6040 WINDOW#1,1,40,1,3:PAPER#1,2:CLS#1:PEN#1,3:FOR a=0 TO 1:LOCATE#1,1,1+a*2:PRI
NT#1,STRING$(40,154);:NEXT:PEN#1,1:LOCATE#1,5,2:PRINT#1,"> S N I D E R ' S M
A Z E <"
6050 WINDOW 1,40,4,24:WINDOW#2,1,40,25,25:PAPER#2,3:CLS#2:PEN#2,2
6060 wd=3:h=1:flag=2:RESTORE 7300:GOSUB 7090
6070 DIM a(0,0),gef(3),suy(3),sux(3),lzeil$(0),gp(15,1,3)
6080 dx=45:DEG:zx=320:zy=172:FOR a=0 TO 2 STEP 2:gp(0,0,a)=262:gp(0,0,a+1)=142:N
EXT:gp(0,0,3)=-142:FOR a=0 TO 3:gp(0,1,a)=-gp(0,0,a):NEXT
6090 FOR b=1 TO 15:FOR a=0 TO 1
6100 gp(b,a,0)=SIN(dx)*zx:gp(b,a,1)=COS(dx)*zy:dx=90+dx:gp(b,a,2)=SIN(dx)*zx:gp(
b,a,3)=COS(dx)*zy
6110 dx=dx+90:NEXT:zx=SIN(45)*zx:zy=SIN(45)*zy:dx=45:NEXT
6120 FOR a=0 TO 3:sux(a)=a-1 AND a<3:suy(3-a)=sux(a):NEXT:su=1
6130 ENT 2,2,-20,1,5,0,2,40,1,1
6140 RETURN
6150 '
7000 '**** Toene ****
7010 '
7020 '=== Bei Lauf gegen die Wand ===
7030 '
7040 ENV 1,3,5,1,1,0,27,15,-1,4
7050 SOUND 1,60,0,0,1,2:SOUND 2,90,0,0,1,2,1:SOUND 4,120,0,0,1,2:RETURN
7060 '
7070 '==== Bei Anfang und Ende ====
7080 '
7090 ENT-1,1,-10,1,1,10,1
7100 READ ton,dauer:t=ton/h:d=wd*dauer*2:vn=1+(vn+1)MOD 15:ev=SGN(ton-1):IF t<0
THEN RETURN
7110 GOSUB 7130:IF flag=2 THEN PEN INT(RND*4):LOCATE 1,1+INT(RND*21):PRINT SPACE
$(1+INT(RND*36))CHR$(144+INT(RND*3))CHR$(18);
7120 GOTO 7090
7130 ENV vn,3*ev,5,1,dauer,0,wd/2,15,-1*ev,ROUND((dauer+1)/3)
7140 SOUND 1,ROUND(t/4),0,0,vn
7150 SOUND 2,t*2,0,0,vn,1
7160 SOUND 4,ROUND(t/2),0,0,vn
7170 RETURN
7180 '=== God Save The Queen ===
7190 DATA 319,4,319,4,284,4
7200 DATA 358,6,319,2,284,4,253,4,253,4,239,4
7210 DATA 253,6,284,2,319,4,284,4,319,4,358,4
7220 DATA 319,8,1,4
7230 DATA 213,4,213,4,213,4,213,6,239,2,253,4
7240 DATA 239,4,239,4,239,4,239,6,253,2,284,4
7250 DATA 253,4,239,2,253,2,284,2,319,2
7260 DATA 253,6,239,2,213,4,190,2,239,2,253,4,284,4
7270 DATA 319,8,1,4,213,4,213,4,213,4,213,6,239,2,253,4,239,4,239,4,239,4,239,6,
253,2,284,4,253,4,239,2,253,2,284,2,319,2,253,6,239,2,213,4,190,2,239,2,253,4,28
4,4,319,16,1,4
7280 DATA-1,0
7290 '=== Home On The Range ===
7300 DATA 426,4,426,4,319,4,284,4,253,8,319,2,358,2,379,4,239,4,239,4
7310 DATA 239,8,253,2,239,2,213,8,319,2,319,2,319,4,358,4,319,4,284,24
7320 DATA 213,12,239,4,253,6,284,2,253,20,426,2,426,2,319,4,319,4,319,4,319,4,35
8,4,319,4
7330 DATA 284,20,426,4,426,4,319,4,284,4,253,8,319,2,358,2,379,4,239,4,239,4
7340 DATA 239,8,239,2,239,2,253,6,284,2,319,4,358,4,319,4,284,4,319,24
7350 DATA -1,0
```